9/17/99

The Deformation of Simply Supported Plates

The eigenfunctions for simply supported rectangular plates are just the product sin(mπx/a)sin(nπy/b). This simple situation makes it pretty easy to find the deformation in terms of series of eigenfunctions. The expression for the deformation converges pretty rapidly and it is practical to use the eigenfunction technique to compute surface shapes. The point of this note is to set up the general procedure and to run through some examples.

The first thing to do is to select the number of eigenfunctions to use

```
STUDENT > N:=3:M:=3:
```

Allocate storage for the eigenfunctions and the coefficients. (These eigenfunctions don't have zero components, but I'd like to keep this general for systems that do.)

```
STUDENT > f:=array(0..M,0..N):A:=array(0..M,0..N):
```

Define the eigenfunctions

```
STUDENT > for m from 0 to M do for n from 0 to N do
          f[m,n]:=sin(m*Pi*x/a)*sin(n*Pi*y/b)
          od:od:
```

Add up the eigenfunctions multiplied by the unknown coefficients to form the truncated representation of the displacement w

```
STUDENT > w:=0:
          for m from 0 to M do for n from 0 to N do
          w:=w+A[m,n]*f[m,n] od:od:
          w;
```

$$A_{1,1}\sin\left(\frac{\pi x}{a}\right)\sin\left(\frac{\pi y}{b}\right)+A_{1,2}\sin\left(\frac{\pi x}{a}\right)\sin\left(2\frac{\pi y}{b}\right)+A_{1,3}\sin\left(\frac{\pi x}{a}\right)\sin\left(3\frac{\pi y}{b}\right)$$
$$+A_{2,1}\sin\left(2\frac{\pi x}{a}\right)\sin\left(\frac{\pi y}{b}\right)+A_{2,2}\sin\left(2\frac{\pi x}{a}\right)\sin\left(2\frac{\pi y}{b}\right)+A_{2,3}\sin\left(2\frac{\pi x}{a}\right)\sin\left(3\frac{\pi y}{b}\right)$$
$$+A_{3,1}\sin\left(3\frac{\pi x}{a}\right)\sin\left(\frac{\pi y}{b}\right)+A_{3,2}\sin\left(3\frac{\pi x}{a}\right)\sin\left(2\frac{\pi y}{b}\right)+A_{3,3}\sin\left(3\frac{\pi x}{a}\right)\sin\left(3\frac{\pi y}{b}\right)$$

form the Laplacian and the biharmonic of w

```
STUDENT > Lw:=diff(diff(w,x),x)+diff(diff(w,y),y);
STUDENT > LLw:=diff(diff(Lw,x),x)+diff(diff(Lw,y),y);
```

Define the surface loading, here taken to be constant

```
STUDENT > stress:=q:
```

Multiply the surface loading by each eigenfunction in turm and integrate over the region to define the coefficients of the truncated expansion of the surface loading

```
STUDENT > s:=array(0..M,0..N):
          for m from 0 to M do for n from 0 to N do
```

```
s[n,m]:=int(int(stress*f[m,n],x=0..a),y=0..b) od:od:
```

Doing the same thing for the expansion of w is tedious and unnecessary. Each term integrates to zero or (a/2)(b/2) depending on its orthogonality, so I take the shortcut of selecting the coefficients of each constant and substituting the integrated value for the integral that would have resulted had I done the entire calculation. (You are welcome to set the thing up correctly and work through it.) As an added side benefit, this procedure puts a nonzero coefficient in front of terms that would otherwise have zeroes and makes the general formula for the coefficients go through without a lot of divide by zero error messages

```
STUDENT > c:=array(0..M,0..N):
          for m from 0 to M do for n from 0 to N do
          factor(coeff(LLw,A[m,n])):
          c[m,n]:=subs(sin(m*Pi*x/a)=a/2,sin(n*Pi*y/b)=b/2,"):
           od:od:
          print(c);
```

array(0 .. 3, 0 .. 3, [

$$(0, 0) = \frac{1}{2}a$$

$$(0, 1) = \frac{1}{2}a$$

$$(0, 2) = \frac{1}{2}a$$

$$(0, 3) = \frac{1}{2}a$$

$$(1, 0) = \frac{1}{2}b$$

$$(1, 1) = \frac{1}{4}\frac{\pi^4 (a^2 + b^2)^2}{a^3 b^3}$$

$$(1, 2) = \frac{1}{4}\frac{\pi^4 (b^2 + 4 a^2)^2}{a^3 b^3}$$

$$(1, 3) = \frac{1}{4}\frac{\pi^4 (b^2 + 9 a^2)^2}{a^3 b^3}$$

$$(2, 0) = \frac{1}{2}b$$

$$(2, 1) = \frac{1}{4}\frac{\pi^4 (4 b^2 + a^2)^2}{a^3 b^3}$$

$$(2, 2) = 4\frac{\pi^4 (a^2 + b^2)^2}{a^3 b^3}$$

$$(2, 3) = \frac{1}{4}\frac{\pi^4 (4 b^2 + 9 a^2)^2}{a^3 b^3}$$

$$(3,0)=\frac{1}{2}b$$

$$(3,1)=\frac{1}{4}\frac{\pi^4(9b^2+a^2)^2}{a^3b^3}$$

$$(3,2)=\frac{1}{4}\frac{\pi^4(9b^2+4a^2)^2}{a^3b^3}$$

$$(3,3)=\frac{81}{4}\frac{\pi^4(a^2+b^2)^2}{a^3b^3}$$

])

Now I can invert the two series representations by equating the coefficients of each term. I print out only the nonzero terms

```
STUDENT > for m from 0 to M do for n from  0 to N do
          A[m,n]:=s[m,n]/c[m,n] od:od:
          for m from 0 to M do for n from 0 to N do
          if(A[m,n]<>0) then print(m,n,A[m,n]) fi od;od;
```

$$1,\ 1,\ 16\frac{b^4a^4q}{\pi^6(a^2+b^2)^2}$$

$$1,\ 3,\ \frac{16}{3}\frac{b^4a^4q}{\pi^6(b^2+9a^2)^2}$$

$$3,\ 1,\ \frac{16}{3}\frac{b^4a^4q}{\pi^6(9b^2+a^2)^2}$$

$$3,\ 3,\ \frac{16}{729}\frac{b^4a^4q}{\pi^6(a^2+b^2)^2}$$

And so we have a four term representation for w

```
STUDENT > eval(w);
```

$$16\frac{b^4a^4q\sin\left(\frac{\pi x}{a}\right)\sin\left(\frac{\pi y}{b}\right)}{\pi^6(a^2+b^2)^2}+\frac{16}{3}\frac{b^4a^4q\sin\left(\frac{\pi x}{a}\right)\sin\left(3\frac{\pi y}{b}\right)}{\pi^6(b^2+9a^2)^2}$$
$$+\frac{16}{3}\frac{b^4a^4q\sin\left(3\frac{\pi x}{a}\right)\sin\left(\frac{\pi y}{b}\right)}{\pi^6(9b^2+a^2)^2}+\frac{16}{729}\frac{b^4a^4q\sin\left(3\frac{\pi x}{a}\right)\sin\left(3\frac{\pi y}{b}\right)}{\pi^6(a^2+b^2)^2}$$

the shape of which we can plot, here for a 1:2 aspect ratio with a pressure loading chosen to make the picture look nice.

```
STUDENT > q:=-50:a:=1:b:=2:
          plot3d(w,x=0..a,y=0..b,axes=BOXED,shading
```

```
=Z,scaling=CONSTRAINED);
q:='q':a:='a':b:='b':
```



To do a different problem I need to redefine the loading and recalculate the coefficients of the truncation. If I leave the order of the truncation alone, that is all I need to do. If I change the order of the truncation, I'll need to redefine everything, essentially starting over from the top of this worksheet. I have done several of these for a square plate following the general pattern below. These are in a separate file accessible from the website..

```
STUDENT > stress:=q*(x/a)^4:

STUDENT > s:=array(0..M,0..N):
          for m from 0 to M do for n from 0 to N do
          s[m,n]:=int(int(stress*f[m,n],x=0..a),y=0..b) od:od:

STUDENT > for m from 0 to M do for n from  0 to N do
          A[m,n]:=s[m,n]/c[m,n] od:od:

STUDENT > q:=1:a:=1:b:=1:plot3d(stress,x=0..a,y=0..b,axes=BOXED,shad
          ing=Z,scaling=UNCONSTRAINED,title=`Load`);q:='q':a:='a':b:
          ='b':q:=-200:a:=1:b:=1:
          plot3d(w,x=0..a,y=0..b,axes=BOXED,shading
          =Z,scaling=UNCONSTRAINED,title=`Deflection`);
          q:='q':a:='a':b:='b':
```






```
STUDENT > save `paradigm.m`:
```